Risoluzione dei sistemi lineari con il metodo di Banachiewicz

Un semplice e efficiente metodo per risolvere i sistemi di equazioni lineari è quello recentemente proposto da Banachiewicz che consente di risolvere sistemi generici simmetrici e non simmetrici anche senza la prevalenza dei termini della diagonale principale sugli altri termini (l'elemento della prima riga e prima colonna non deve essere nullo). Inoltre il metodo non ha il difetto di aggravare le approssimazioni con cui si eseguono i calcoli.

Il metodo è fondato sulla teoria dei cracoviani. Un cracoviano è una generica matrice rettangolare. Una breve esposizione del metodo è riportato in "Odone Belluzzi: Scienza delle Costruzioni, volume II, Zanichelli Bologna, pag. 287". Per un maggiore approfondimento vedere "T. Banachiewicz: Etudes d'analyse pratique, «Acc. Polacca di Sc. e Lett.» , Cracovia, 1938".

1) Calcolo della matrice inversa

Dato il sistema di n equazioni lineari,

$$A\,X = I$$

dove I è la matrice unità, la matrice inversa A^{-1} di A tale per cui risulta

$$X = A^{-1}\,I = A^{-1}$$

si può ottenere, come noto, dalla risoluzione dello stesso sistema lineare.
Allo scopo si procede secondo il metodo di Banachiewicz. Si scompone la matrice A in due matrici triangolari superiori V e W (V e W cracoviani canonici con W a diagonale unitaria) tali che risulti verificata l'uguaglianza del seguente prodotto cracoviano

$$A = W \bullet V$$

da cui la soluzione, è ottenuta dal sistema

$$W\,X = D$$

con una facile sostituzione all'indietro essendo W matrice triangolare superiore.

E' definito prodotto di due cracoviani W●V un cracoviano A il cui generico elemento vale

$$a_{ij} = \sum_{k=1}^{n} w_{kj} v_{ki}$$

cioè un prodotto colonna per colonna.

I termini delle matrici V e W (primo e secondo cracoviano) sono dati dalle formule:

$$v_{ij} = a_{ji} - \sum_{k=1}^{i-1} w_{ki} v_{kj} \qquad 1)$$

$$w_{ij} = (a_{ij} - \sum_{k=1}^{i-1} w_{kj} v_{ki}) / v_{ii} \qquad 2)$$

i termini della matrice D (cracoviano dei termini noti) sono dati dalla formula

$$d_{ij} = (i_{ij} - \sum_{k=1}^{i-1} d_{kj} v_{ki}) / v_{ii} \qquad 3)$$

i termini della matrice X sono dati dalla formula

$$x_{ij} = (d_{ij} - \sum_{k=i+1}^{n} w_{ik} x_{kj}) / w_{ii} = d_{ij} - \sum_{k=i+1}^{n} w_{ik} x_{kj} \qquad 4)$$

con sostituzione all'indietro, cioè con i da n a 1.

Con alcune semplici considerazioni è possibile basarsi sul solo spazio di memoria occupato dalla matrice A(n,n) e da un vettore aggiuntivo di n elementi. Infatti, considerando anziché la matrice V triangolare superiore la sua trasposta U triangolare inferiore, cioè con $\tilde{U}=V$, le 1) e 2) diventano

$$u_{ji} = a_{ji} - \sum_{k=1}^{i-1} w_{ki} u_{jk} \qquad 1')$$

$$w_{ij} = (a_{ij} - \sum_{k=1}^{i-1} w_{kj} u_{ik}) / u_{ii} \qquad 2')$$

ed osservando che i termini della matrice A sono usati una sola volta e per il calcolo del corrispondente termine delle matrici W e U, e che l'elemento calcolato viene posto nella stessa posizione dell'elemento rilasciato, le due formule 1') e 2') possono scriversi formalmente con riferimento alla sola matrice **A**

$$a_{ji} = a_{ji} - \sum_{k=1}^{i-1} a_{ki} a_{jk} \qquad \text{con } j \geq i \quad 1'')$$

$$a_{ij} = (a_{ij} - \sum_{k=1}^{i-1} a_{kj} a_{ik}) / a_{ii} \qquad \text{con } j > i \qquad 2'')$$

con le operazioni da eseguire nella giusta sequenza. Il cracoviano V è ora posto nella parte triangolare inferiore di A mentre il cracoviano W è ora posto nella parte triangolare superiore di A a meno della sua diagonale i cui termini sono di valore unitario e non memorizzati.
Essendo poi la I matrice identità, cioè risulta

$I_{ij} = 0$ per $i \neq j$
$I_{ij} = 1$ per $i = j$

dalla 3) si deduce che la matrice D è triangolare inferiore, cioè risulta

$d_{ij} = 0$ per $i < j$
$d_{ij} \neq 0$ per $i \geq j$

quindi la 3) può scriversi

$$d_{ij} = (i_{ij} - \sum_{k=j}^{i-1} d_{kj} v_{ki}) / v_{ii} = (i_{ij} - \sum_{k=j}^{i-1} d_{kj} u_{ik}) / u_{ii} \qquad 3')$$

dove si osserva che il calcolo dell'elemento d_{ij} richiede l'uso dei soli elementi u_{ik} posti sulla stessa riga e con $k \geq j$. Quindi gli elementi d_{ij} possono mettersi nella matrice U e dunque in A. La 3) può dunque scriversi formalmente con riferimento alla sola matrice A

$$a_{ij} = (i_{ij} - \sum_{k=j}^{i-1} a_{kj} a_{ik}) / a_{ii} \qquad \text{con } i \geq j \qquad 3'')$$

Infine, essendo D matrice triangolare inferiore, le 4) si scrivono:

$$x_{ij} = d_{ij} - \sum_{k=i+1}^{n} w_{ik} x_{kj} \qquad \text{per } i \geq j$$

$$x_{ij} = 0 - \sum_{k=i+1}^{n} w_{ik} x_{kj} \qquad \text{per } i < j$$

che tenendo conto delle posizioni precedenti si trasformano nelle seguenti

$$x_{ij} = a_{ij} - \sum_{k=i+1}^{n} a_{ik} x_{kj} \qquad \text{per } i \geq j$$

$$x_{ij} = 0 - \sum_{k=i+1}^{n} a_{ik} x_{kj} \qquad \text{per } i < j$$

da cui tenendo conto che il calcolo di una riga di **X** servono solo gli elementi di A della stessa riga, gli elementi x_{ij} possono mettersi in A_{ij} ogni volta che tutta la riga è calcolata nella sostituzione all'indietro. Formalmente si può scrivere

$$a_{ij} = a_{ij} - \sum_{k=i+1}^{n} a_{ik} a_{kj} \qquad \text{per } i \geq j,\ i = n,1$$

$$a_{ij} = 0 - \sum_{k=i+1}^{n} a_{ik} a_{kj} \qquad \text{per } i < j,\ i = n,1$$

con l'avvertenza di sostituire X in A riga per riga.

2) Risoluzione sistema lineare

Dato il generico sistema di n equazioni lineari, A X = B

dove A è la matrice dei coefficienti, B il vettore dei termini noti ed X il vettore delle incognite, si scompone la matrice A in due matrici triangolari superiori V e W (V e W cracoviani canonici con W a diagonale unitaria) tali che risulti verificata l'uguaglianza del seguente prodotto cracoviano

A = W ● V

da cui la soluzione, è ottenuta dal sistema

W X = B

con una facile sostituzione all'indietro essendo W matrice triangolare superiore.
I termini delle matrici V e W (primo e secondo cracoviano) sono dati dalle formule 1) e 2), valgono ovviamente anche le 1") e 2").
I termini del vettore D (vettore dei termini noti) sono dati dalla formula

$$d_i = (b_i - \sum_{k=1}^{i-1} d_k v_{ki}) / v_{ii} \qquad \text{3a)}$$

i termini della matrice X sono dati dalla formula

$$x_i = (d_i - \sum_{k=i+1}^{n} w_{ik} x_k) / w_{ii} = d_i - \sum_{k=i+1}^{n} w_{ik} x_k \qquad \text{4a)}$$

con sostituzione all'indietro, cioè con i da n a 1.
La 3a) può scriversi

$$d_i = (b_i - \sum_{k=1}^{i-1} d_k v_{ki}) / v_{ii} = (b_i - \sum_{k=1}^{i-1} d_k u_{ik}) / u_{ii} \qquad \text{3'a)}$$

dove si osserva che il calcolo dell'elemento d_i richiede l'uso dei soli elementi d_k con $k<i$. Quindi gli elementi d_i possono mettersi nel vettore B. La 3'a) può dunque scriversi formalmente con riferimento alla matrice A

$$b_i = (b_i - \sum_{k=1}^{i-1} b_k a_{ik}) / a_{ii} \qquad \text{3"a)}$$

La 4a) può infine scriversi

$$x_i = b_i - \sum_{k=i+1}^{n} a_{ik} x_k$$

3) Risoluzione sistema lineare simmetrico

Per sistemi lineari simmetrici, come è facile costatare, risulta

$w_{ij}=v_{ij}/v_{ii}$

e le formule 1) e 4a) diventano

$$v_{ij} = a_{ji} - \sum_{k=1}^{i-1} v_{ki} v_{kj} / v_{kk} \qquad \text{1b)}$$

$$x_i = d_i - \sum_{k=i+1}^{n} w_{ik} x_k = (d_i v_{ii} - \sum_{k=i+1}^{n} v_{ik} x_k) / v_{ii} \qquad \text{4b)}$$

mentre la 3a) rimane valida, dunque anche la 3"a). Tenendo conto della simmetria della matrice A ed inserendo la matrice V in A senza trasporla (prende il posto della matrice W) ed inserendo X in B, le formule possono scriversi formalmente con riferimento alla matrice A

$$a_{ij} = a_{ij} - \sum_{k=1}^{i-1} a_{ki} a_{kj} / a_{kk} \qquad \text{per } j \geq i$$

$$b_i = (b_i - \sum_{k=1}^{i-1} b_k a_{ki}) / a_{ii}$$

$$b_i = (b_i a_{ii} - \sum_{k=i+1}^{n} a_{ik} b_k) / a_{ii}$$

4) Risoluzione sistema lineare simmetrico a banda

Il metodo di Banachiewicz agevola la memorizzazione a banda della matrice delle incognite avendo la particolare caratteristica che la larghezza di banda delle righe della matrice A durante la risoluzione del sistema non va ad occupare posizioni oltre quelle iniziali purchè si consideri per ogni riga anche i termini nulli oltre l'ultimo non nullo in modo tale che la larghezza di banda della riga i+1 non risulti inferiore alla larghezza di banda della riga i diminuita di 1.

5) Listati

Di seguito sono riportati i listati delle routines in Fortran. Si ricorda che in fortran le matrici sono memorizzate a colonne.

```
SUBROUTINE BanachiewiczInv (a,n,Err)
!
! Subroutine FORTRAN
!
! Inverte matrice a(n,n) con il metodo di Banachiewicz
! ed il risultato è riposto in a(n,n).
!
! a(n,n): matrice coefficienti (IN), matrice inversa (OUT)
! n     : numero equazioni
! Err   : =0, nessun errore
!         >0, a(i,i) nullo
!
INTEGER(4) n,Err,i,j,k
REAL(8) a(n,n),b(n),r

Err=0

! Calcolo primo e secondo cracoviano

do i=1,n
  do j=i,n
    do k=1,i-1
      a(j,i)=a(j,i)-a(k,i)*a(j,k)
    end do
  end do
  if (a(i,i).eq.0d0) then; Err=i; return; end if
  do j=i+1,n
    do k=1,i-1
      a(i,j)=a(i,j)-a(k,j)*a(i,k)
    end do
    a(i,j)=a(i,j)/a(i,i)
  end do
end do

! Calcolo cracoviano "termini noti"

do j=1,n
  r=1d0
  do i=j,n
    do k=j,i-1
      r=r-a(k,j)*a(i,k)
    end do
     a(i,j)=r/a(i,i)
    r=0d0
  end do
end do
```

```
! Calcolo incognite

do i=n,1,-1
  do j=1,n
    b(j)=0d0
    if (j.le.i) b(j)=a(i,j)
    do k=i+1,n
      b(j)=b(j)-a(i,k)*a(k,j)
    end do
  end do
  do k=1,n
    a(i,k)=b(k)
  end do
end do

END SUBROUTINE

SUBROUTINE BanachiewiczSist (A,X,B,n,Err)
!
! Subroutine FORTRAN
!
! Risoluzione sistema lineare A*X=B
!
! A(n,n): matrice dei coefficienti
! X(n)  : vettori delle incognite
! B(n)  : vettori termini noti
! n     : numero equazioni
! Err   : =0, nessun errore
!         >0, a(i,i) nullo
!
INTEGER(4) n,Err,i,j,k
REAL(8) A(n,n),X(n),B(n),r

Err=0

! Calcolo primo e secondo cracoviano

DO i=1,n
  DO j=i,n
    DO k=1,i-1
      A(j,i)=A(j,i)-A(k,i)*A(j,k)
    END DO
  END DO
  IF (A(i,i).EQ.0d0) THEN; Err=i; RETURN; END IF
  DO j=i+1,n
    DO k=1,i-1
      A(i,j)=A(i,j)-A(k,j)*A(i,k)
    END DO
    A(i,j)=A(i,j)/A(i,i)
  END DO
END DO

! Calcolo cracoviano termini noti

DO i=1,n
  r=B(i)
```

```
  DO k=1,i-1
    r=r-B(k)*A(i,k)
  END DO
  B(i)=r/A(i,i)
END DO

! Calcolo incognite

DO i=n,1,-1
  X(i)=B(i)
  DO k=i+1,n
    X(i)=X(i)-A(i,k)*X(k)
  END DO
END DO

END SUBROUTINE

SUBROUTINE BanachiewiczSistSimm (a,b,n,Err)
!
! Subroutine FORTRAN
!
! Risoluzione sistema lineare simmetrico A*X=B
!
! a(n,n): matrice dei coefficienti
! b(n)  : vettori termini noti (IN), vettore delle incognite (OUT)
! n     : numero equazioni
! Err   : =0, nessun errore
!         >0, a(i,i) nullo
!
INTEGER(4) n,Err,i,j,k
REAL(8) a(n,n),b(n),r

Err=0

IF (a(1,1).LE.0d0) THEN; Err=1; RETURN; END IF

DO i=1,n
  DO k=1,i-1
    r=a(k,i)/a(k,k)
    DO j=i,n
      a(i,j)=a(i,j)-r*a(k,j)
    END DO
  END DO
  IF (a(i,i).LE.0d0) THEN; Err=i; RETURN; END IF
END DO

DO i=1,n
  r=b(i)
  DO k=1,i-1
    r=r-b(k)*a(k,i)
  END DO
  b(i)=r/a(i,i)
END DO

DO i=n,1,-1
  r=b(i)*a(i,i)
  DO k=i+1,n
```

```
    r=r-a(i,k)*b(k)
  END DO
  b(i)=r/a(i,i)
END DO

END SUBROUTINE
```

6) Matrici note

Matrice A

$$
A = \begin{vmatrix}
n & n-1 & n-2 & \dots & 3 & 2 & 1 \\
n-1 & n & n-1 & \dots & 4 & 3 & 2 \\
\dots & \dots & \dots & \dots & \dots & \dots & \dots \\
\dots & \dots & \dots & \dots & \dots & \dots & \dots \\
\dots & \dots & \dots & \dots & \dots & \dots & \dots \\
1 & 2 & 3 & \dots & n-2 & n-1 & n
\end{vmatrix}
$$

$$
A^{-1} = \begin{vmatrix}
a & -1/2 & 0 & & & & b \\
-1/2 & 1 & -1/2 & 0 & & & 0 \\
0 & -1/2 & 1 & -1/2 & 0 & \dots & 0 \\
\dots & \dots & \dots & \dots & \dots & \dots & \dots \\
\dots & \dots & \dots & \dots & \dots & \dots & \dots \\
b & 0 & \dots & \dots & \dots & -1/2 & a
\end{vmatrix}
$$

Con a=(n+2)/(2n+2) e b=1/(2n+2)